The tensorial double at $\epsilon^2 = 0$

In[∘]:=
```
ϵ /: ϵ^d_· /; d > 1 := 0;
```

In[∘]:=
```
CF[ℰ_] := ExpandDenominator@
   ExpandNumerator@Together[Expand[ℰ] //. e^x_ e^y_ :→ e^(x+y) /. e^x_ :→ e^CF[x]];
Kδ /: Kδ_{i_,j_} := If[i === j, 1, 0];
```

Zip and Bind

In[∘]:=
```
𝔼 /: 𝔼[L1_, Q1_, P1_] ≡ 𝔼[L2_, Q2_, P2_] :=
  CF[L1 == L2] ∧ CF[Q1 == Q2] ∧ CF[Normal[P1 - P2] == 0];
𝔼 /: 𝔼[L1_, Q1_, P1_] 𝔼[L2_, Q2_, P2_] := 𝔼[L1 + L2, Q1 + Q2, P1 * P2];
```

In[∘]:=
```
{t*, b*, y*, a*, x*, z*} = {τ, β, η, α, ξ, ζ};
{τ*, β*, η*, α*, ξ*, ζ*} = {t, b, y, a, x, z}; (u_i_)* := (u*)_i;
```

In[∘]:=
```
Zip_{}[P_] := P; Zip_{ζ_,ζs___}[P_] := (Expand[P // Zip_{ζs}] /. f_. ζ^d_· :→ ∂_{ζ*,d} f) /. ζ* → 0
```

QZip implements the “Q-level zips” on $\mathbb{E}(L, Q, P) = Pe^{L+Q}$. Such zips regard the L variables as scalars.

In[∘]:=
```
QZip_{ζs_List,simp_}@𝔼[L_, Q_, P_] := Module[{ζ, z, zs, c, ys, ηs, qt, zrule, Q1, Q2},
   zs = Table[ζ*, {ζ, ζs}];
   c = Q /. Alternatives @@ (ζs ⋃ zs) → 0;
   ys = Table[∂_ζ (Q /. Alternatives @@ zs → 0), {ζ, ζs}];
   ηs = Table[∂_z (Q /. Alternatives @@ ζs → 0), {z, zs}];
   qt = Inverse@Table[Kδ_{z,ζ*} - ∂_{z,ζ}Q, {ζ, ζs}, {z, zs}];
   zrule = Thread[zs → qt.(zs + ys)];
   Q2 = (Q1 = c + ηs.zs /. zrule) /. Alternatives @@ zs → 0;
   simp /@ 𝔼[L, Q2, Det[qt] e^-Q2 Zip_ζs[e^Q1 (P /. zrule)]] ];
QZip_{ζs_List} := QZip_{ζs,CF};
```

LZip implements the “L-level zips” on $\mathbb{E}(L, Q, P) = Pe^{L+Q}$. Such zips regard all of Pe^Q as a single”P”. Here the z’s are b and α and the ζ’s are β and a.

In[ ]:=
```
LZip_{ξs_List,simp_}@𝔼[L_, Q_, P_] :=
  Module[{ξ, z, zs, c, ys, ηs, ηsrule, lt, zrule, L1, L2, Q1, Q2},
   (*Print["LZipping"];*)
   (*ξs//Echo;*)
   zs = Table[ξ*, {ξ, ξs}];
   c = L /. Alternatives @@ (ξs ⋃ zs) → 0;
   ys = Table[∂_ξ (L /. Alternatives @@ zs → 0), {ξ, ξs}];
   ηs = Table[∂_z (L /. Alternatives @@ ξs → 0), {z, zs}];
   ηsrule = Table[z* -> ∂_z (L /. Alternatives @@ ξs → 0), {z, zs}]; (*NEW*)
   lt = Inverse@Table[Kδ_{z,ξ*} - ∂_{z,ξ} L, {ξ, ξs}, {z, zs}];
   zrule = Thread[zs → lt.(zs + ys)];
   L1 = c + ηs.zs /. zrule;
   L2 = L1 /. Alternatives @@ zs → 0;
   Q1 = Q /. Join[zrule, ηsrule];
   Q2 = Q1 /. Join[{Alternatives @@ zs → 0}, ηsrule];
   (*{Det[lt] e^(-L2-Q2), Zip_ξs[e^(L1+Q1) (P/.zrule)]}//Echo;*)
   simp /@ 𝔼[L2, Q2, Det[lt] e^(-L2-Q2) Zip_ξs[e^(L1+Q1) (P /. zrule)]] ];
LZip_{ξs_List} := LZip_{ξs,CF};
```

In[ ]:=
```
Bind_{}[L_, R_] := L R;
Bind_{is__}[L_𝔼, R_𝔼] := Module[{n},
  Times[
    L /. Table[(v : t | b | a | x | y)_i → v_{n@i}, {i, {is}}],
    R /. Table[(v : τ | β | α | ξ | η)_i → v_{n@i}, {i, {is}}]
   ] // LZip_{Flatten@Table[{τ_{n@i},β_{n@i},a_{n@i}},{i,{is}}]} // QZip_{Flatten@Table[{ξ_{n@i},y_{n@i}},{i,{is}}]} ];
B_{l_List} := Bind_l; B_{is___} := Bind_{is};
```

The two halves

In[ ]:=
```
(*Hopf algebra on the a,x side*)
tam_{i_,j_->k_} := 𝔼[(α_i + α_j) a_k, (e^(-α_j) ξ_i + ξ_j) x_k, 1]
taΔ_{i_->j_,k_} := 𝔼[α_i (a_j + a_k), ξ_i (x_j + x_k), 1 + ϵ ξ_i x_k (-a_j + 1/2 ξ_i x_j)]
taS_{i_} := 𝔼[-α_i a_i, - e^(α_i) ξ_i x_i, 1 - ϵ e^(α_i) ξ_i x_i (a_i + 1/2 e^(α_i) ξ_i x_i)]
taSi_{i_} := 𝔼[-α_i a_i, -e^(α_i) ξ_i x_i, 1 - ϵ e^(α_i) ξ_i x_i (a_i - 1 + 1/2 e^(α_i) ξ_i x_i)]
(*Hopf algebra on the y,b side*)
tbm_{i_,j_->k_} := 𝔼[(β_i + β_j) b_k, (η_i + η_j) y_k, 1 - ϵ η_j y_k β_i]
tbΔ_{i_->j_,k_} := 𝔼[β_i (b_j + b_k), η_i (e^(-b_k) y_j + y_k), 1 + 1/2 ϵ η_i^2 y_j y_k e^(-b_k)]
tbS_{i_} := 𝔼[-β_i b_i, -e^(b_i) η_i y_i, 1 - ϵ e^(b_i) η_i y_i (β_i + 1/2 e^(b_i) η_i y_i)]
tbSi_{i_} := 𝔼[-β_i b_i, -e^(b_i) η_i y_i, 1 - ϵ e^(b_i) η_i y_i (β_i - 1 + 1/2 e^(b_i) η_i y_i)]
```

First check that on the generators this agrees with our conventions in SLPortfolio.pdf with $\hbar = \gamma = 1$

```
In[∘]:= {
   "[a,x] = " ((𝔼[0, 0, a₂ x₁] ~B₁,₂~ tam₁,₂→₁)⟦3⟧ - (𝔼[0, 0, a₁ x₂] ~B₁,₂~ tam₁,₂→₁)⟦3⟧),
   "[b,y] = " ((𝔼[0, 0, y₂ b₁] ~B₁,₂~ tbm₁,₂→₁)⟦3⟧ - (𝔼[0, 0, y₁ b₂] ~B₁,₂~ tbm₁,₂→₁)⟦3⟧)
  } /. {ⅇ → E, z_1 → z} // Simplify
{
   " Δ(a) = " ((𝔼[0, 0, a₁] ~B₁~ taΔ₁→₁,₂)⟦3⟧),
   " Δ(x) = " ((𝔼[0, 0, x₁] ~B₁~ taΔ₁→₁,₂)⟦3⟧),
   " Δ(b) = " ((𝔼[0, 0, b₁] ~B₁~ tbΔ₁→₁,₂)⟦3⟧),
   " Δ(y) = " ((𝔼[0, 0, y₁] ~B₁~ tbΔ₁→₁,₂)⟦3⟧)
  } /. {ⅇ → E} // Simplify
{
   " S(a) = " ((𝔼[0, 0, a₁] ~B₁~ taS₁)⟦3⟧),
   " S(x) = " ((𝔼[0, 0, x₁] ~B₁~ taS₁)⟦3⟧),
   " S(b) = " ((𝔼[0, 0, b₁] ~B₁~ tbS₁)⟦3⟧),
   " S(y) = " ((𝔼[0, 0, y₁] ~B₁~ tbS₁)⟦3⟧)
  } /. {ⅇ → E, z_1 → z} // Simplify
```

Out[∘]= $\{-[a,x] = x,\ -[b,y] = y\,\epsilon\}$

Out[∘]= $\{\ \Delta(a) = (a_1 + a_2),\ \Delta(x) = (x_1 + (1 - \epsilon\, a_1)\, x_2),\ \Delta(b) = (b_1 + b_2),\ \Delta(y) = (e^{-b_2} y_1 + y_2)\}$

Out[∘]= $\{-\ S(a) = a,\ -\ S(x) = (x + a\, x\, \epsilon),\ -\ S(b) = b,\ -\ S(y) = e^{b}\, y\}$

Hopf algebra axioms on both sides separately

(co)-associativity on both sides

```
In[∘]:= {(taΔ₁→₁,₂ ~B₂~ taΔ₂→₂,₃) ≡ (taΔ₁→₁,₃ ~B₁~ taΔ₁→₁,₂),
  (tbΔ₁→₁,₂ ~B₂~ tbΔ₂→₂,₃) ≡ (tbΔ₁→₁,₃ ~B₁~ tbΔ₁→₁,₂),
  (tam₁,₂→₁ ~B₁~ tam₁,₃→₁) ≡ (tam₂,₃→₂ ~B₂~ tam₁,₂->₁),
  (tbm₁,₂→₁ ~B₁~ tbm₁,₃→₁) ≡ (tbm₂,₃→₂ ~B₂~ tbm₁,₂->₁)}
```

Out[∘]= {True, True, True, True}

Δ is an algebra morphism

```
In[∘]:= {tam₁,₂→₁ ~B₁~ taΔ₁→₁,₂ ≡ (taΔ₁→₁,₃ taΔ₂→₂,₄) ~B₁,₂,₃,₄~ (tam₃,₄→₂ tam₁,₂→₁),
  tbm₁,₂→₁ ~B₁~ tbΔ₁→₁,₂ ≡ (tbΔ₁→₁,₃ tbΔ₂→₂,₄) ~B₁,₂,₃,₄~ (tbm₃,₄→₂ tbm₁,₂→₁)} /. ⅇ → E
```

Out[∘]= {True, True}

S is convolution inverse of id

```
In[∘]:= {(taΔ₁→₁,₂ ~B₁~ taS₁) ~B₁,₂~ tam₁,₂→₁, (taΔ₁→₁,₂ ~B₂~ taS₂) ~B₁,₂~ tam₁,₂→₁} /. ⅇ → E
{(tbΔ₁→₁,₂ ~B₁~ tbS₁) ~B₁,₂~ tbm₁,₂→₁, (tbΔ₁→₁,₂ ~B₂~ tbS₂) ~B₁,₂~ tbm₁,₂→₁} /. ⅇ → E
```

Out[∘]= {𝔼[0, 0, 1], 𝔼[0, 0, 1]}

Out[∘]= {𝔼[0, 0, 1], 𝔼[0, 0, 1]}

Si is the inverse of S

```
In[∘]:= {taSi₁~B₁~taS₁ ≡ 𝔼[a₁ α₁, x₁ ξ₁, 1], taS₁~B₁~taSi₁ ≡ 𝔼[a₁ α₁, x₁ ξ₁, 1]} /. 𝕖 → E
        {tbSi₁~B₁~tbS₁ ≡ 𝔼[b₁ β₁, y₁ η₁, 1], tbS₁~B₁~tbSi₁ ≡ 𝔼[b₁ β₁, y₁ η₁, 1]} /. 𝕖 → E
Out[∘]= {True, True}
Out[∘]= {True, True}
```

S is an algebra anti-(co)morphism

```
In[∘]:= {tam₁,₂→₁~B₁~taS₁ ≡ (taS₁ taS₂)~B₁,₂~tam₂,₁→₁,
          tbm₁,₂→₁~B₁~tbS₁ ≡ (tbS₁ tbS₂)~B₁,₂~tbm₂,₁→₁} /. 𝕖 → E
        {taS₁~B₁~taΔ₁→₁,₂ ≡ taΔ₁→₂,₁~B₁,₂~(taS₁ taS₂),
          tbS₁~B₁~tbΔ₁→₁,₂ ≡ tbΔ₁→₂,₁~B₁,₂~(tbS₁ tbS₂)} /. 𝕖 → E
Out[∘]= {True, True}
Out[∘]= {True, True}
```

Pairing

```
In[∘]:= tP_{i_,j_} := 𝔼[β_i α_j, η_i ξ_j, 1 + 1/4 ϵ η_i^2 ξ_j^2]
```

```
In[∘]:= qfac[k_, q_] := (1 - q)^-k QPochhammer[q, q, k] // FunctionExpand
        qfe[k_] := Normal[Series[qfac[k, E^ρ], {ρ, 0, 1}]] /. {ρ → ϵ}
        Table[𝔼[0, 0, y₁^r b₁^s a₂^t x₂^u]~B₁,₂~tP₁,₂ ≡ 𝔼[0, 0, Kδ_{r,u} Kδ_{s,t} qfe[r] s!],
          {r, 0, 4}, {s, 0, 4}, {t, 0, 4}, {u, 0, 4}] // Flatten // Union
Out[∘]= {True}
```

Pairing axioms

```
In[∘]:= {(tbm₁,₂→₁ 𝔼[α₃ a₃, ξ₃ x₃, 1])~B₁,₃~tP₁,₃ ≡
          (𝔼[β₁ b₁, η₁ y₁, 1] 𝔼[β₂ b₂, η₂ y₂, 1] taΔ₃→₄,₅)~B₁,₄~tP₁,₄~B₂,₅~tP₂,₅
         , (tbΔ₁→₁,₂ 𝔼[α₃ a₃, ξ₃ x₃, 1] 𝔼[α₄ a₄, ξ₄ x₄, 1])~B₁,₃~tP₁,₃~B₂,₄~tP₂,₄ ≡
          (𝔼[β₁ b₁, η₁ y₁, 1] tam₃,₄→₃)~B₁,₃~tP₁,₃ }
Out[∘]= {True, True}
```

```
In[∘]:= {(tbS₁ 𝔼[α₂ a₂, ξ₂ x₂, 1])~B₁,₂~tP₁,₂ ≡ (𝔼[β₁ b₁, η₁ y₁, 1] taS₂)~B₁,₂~tP₁,₂,
         (tbSi₁ 𝔼[α₂ a₂, ξ₂ x₂, 1])~B₁,₂~tP₁,₂ ≡ (𝔼[β₁ b₁, η₁ y₁, 1] taSi₂)~B₁,₂~tP₁,₂}
Out[∘]= {True, True}
```

The Double

The double multiplication (should really bind the a’s and b’s separately)

```
tdm_{i_,j_→k_} :=
  tdm_{i,j→k} = Simplify /@ Expand /@ ((𝔼[β_i b_i + α_j a_j, η_i y_i + ξ_j x_j, 1] (taΔ_{i→h1,h2}~B_{h2}~taΔ_{h2→h2,h3})
        (tbΔ_{j→t1,t2}~B_{t2}~tbΔ_{t2→t2,t3}))~B_{h3}~taSi_{h3}~B_{t1,h3}~(tP_{t1,h3})~
      B_{t3,h1}~(tP_{t3,h1})~B_{h2,j,i,t2}~(tam_{h2,j→k} tbm_{i,t2→-k}) /. {𝕖 → E, u_{_-k} :> u_k})
```

$\mathrm{tdm}_{i,j\to k}$

Out[∘]= $\mathbb{E}\Big[a_k(\alpha_i+\alpha_j)+b_k(\beta_i+\beta_j),\ y_k\left(\eta_i+e^{-\alpha_i}\eta_j\right)+e^{-b_k}\left(-1+e^{b_k}\right)\eta_j\xi_i+x_k\left(e^{-\alpha_j}\xi_i+\xi_j\right),$
$\frac{1}{4}e^{-2b_k-\alpha_i-\alpha_j}\Big(2e^{b_k}\epsilon\, y_k\eta_j\left(-2e^{b_k+\alpha_j}\beta_i+\left(2e^{b_k}x_k+e^{\alpha_j}\left(-3+e^{b_k}\right)\eta_j\right)\xi_i\right)+$
$e^{\alpha_i}\Big(2e^{b_k}\epsilon\, x_k\xi_i\left(-2e^{b_k}\beta_j+\left(-3+e^{b_k}\right)\eta_j\xi_i\right)+$
$e^{\alpha_j}\left(4e^{2b_k}+4e^{b_k}\epsilon\, a_k\eta_j\xi_i+\left(3-4e^{b_k}+e^{2b_k}\right)\epsilon\,\eta_j^2\xi_i^2\right)\Big)\Big)\Big]$

```
(*Deriving tdS using tdm*)
tdS_i_ := tdS_i = ((tbSi_1 taS_2) ~B_{1,2}~ tdm_{2,1→i}) /. {𝕖 → E, z_{_1|2} → z_i}
```

tdS_i

Out[∘]= $\mathbb{E}\Big[-a_i\alpha_i-b_i\beta_i,\ e^{-b_i}\left(-e^{2b_i+\alpha_i}y_i\eta_i-e^{b_i+\alpha_i}x_i\xi_i-e^{b_i+\alpha_i}\eta_i\xi_i+e^{2b_i+\alpha_i}\eta_i\xi_i\right),$
$\frac{1}{4}e^{-2b_i}\Big(4e^{2b_i}+4e^{3b_i+\alpha_i}\epsilon\, y_i\eta_i-4e^{3b_i+\alpha_i}\epsilon\, y_i\beta_i\eta_i-2e^{4b_i+2\alpha_i}\epsilon\, y_i^2\eta_i^2-$
$4e^{2b_i+\alpha_i}\epsilon\, a_i x_i\xi_i-4e^{2b_i+\alpha_i}\epsilon\, x_i\beta_i\xi_i+4e^{2b_i+\alpha_i}\epsilon\,\eta_i\xi_i-4e^{3b_i+\alpha_i}\epsilon\,\eta_i\xi_i+$
$4e^{3b_i+\alpha_i}\epsilon\, a_i\eta_i\xi_i-4e^{3b_i+2\alpha_i}\epsilon\, x_i y_i\eta_i\xi_i-4e^{2b_i+\alpha_i}\epsilon\,\beta_i\eta_i\xi_i+4e^{3b_i+\alpha_i}\epsilon\,\beta_i\eta_i\xi_i-$
$2e^{3b_i+2\alpha_i}\epsilon\, y_i\eta_i^2\xi_i+6e^{4b_i+2\alpha_i}\epsilon\, y_i\eta_i^2\xi_i-2e^{2b_i+2\alpha_i}\epsilon\, x_i^2\xi_i^2-2e^{2b_i+2\alpha_i}\epsilon\, x_i\eta_i\xi_i^2+$
$6e^{3b_i+2\alpha_i}\epsilon\, x_i\eta_i\xi_i^2-e^{2b_i+2\alpha_i}\epsilon\,\eta_i^2\xi_i^2+4e^{3b_i+2\alpha_i}\epsilon\,\eta_i^2\xi_i^2-3e^{4b_i+2\alpha_i}\epsilon\,\eta_i^2\xi_i^2\Big)\Big]$

In[∘]:=
```
(*Deriving tdΔ using tdm*)
(tbΔ_{i→3,1} taΔ_{i→2,4}) ~B_{1,2,3,4}~ (tdm_{3,4→k} tdm_{1,2→j})
```

Out[∘]= $\mathbb{E}\Big[a_j\alpha_i+a_k\alpha_i+b_j\beta_i+b_k\beta_i,\ e^{-b_j}\left(e^{b_j}y_j\eta_i+y_k\eta_i+e^{b_j}x_j\xi_i+e^{b_j}x_k\xi_i\right),$
$\frac{1}{2}e^{-b_j}\left(2e^{b_j}+\epsilon\, y_j y_k\eta_i^2-2e^{b_j}\epsilon\, a_j x_k\xi_i+e^{b_j}\epsilon\, x_j x_k\xi_i^2\right)\Big]$

In[∘]:=

$\mathrm{tdm}_{i\_,j\_\to k\_} := \mathbb{E}\Big[a_k(\alpha_i+\alpha_j)+b_k(\beta_i+\beta_j),\ y_k\left(\eta_i+e^{-\alpha_i}\eta_j\right)+e^{-b_k}\left(-1+e^{b_k}\right)\eta_j\xi_i+x_k\left(e^{-\alpha_j}\xi_i+\xi_j\right),$
$\frac{1}{4}e^{-2b_k-\alpha_i-\alpha_j}\Big(2e^{b_k}\epsilon\, y_k\eta_j\left(-2e^{b_k+\alpha_j}\beta_i+\left(2e^{b_k}x_k+e^{\alpha_j}\left(-3+e^{b_k}\right)\eta_j\right)\xi_i\right)+e^{\alpha_i}\Big(2e^{b_k}\epsilon\, x_k\xi_i$
$\left(-2e^{b_k}\beta_j+\left(-3+e^{b_k}\right)\eta_j\xi_i\right)+e^{\alpha_j}\left(4e^{2b_k}+4e^{b_k}\epsilon\, a_k\eta_j\xi_i+\left(3-4e^{b_k}+e^{2b_k}\right)\epsilon\,\eta_j^2\xi_i^2\right)\Big)\Big)\Big]$

$\mathrm{td}\Delta_{i\_\to j\_,k\_} := \mathbb{E}\Big[a_j\alpha_i+a_k\alpha_i+b_j\beta_i+b_k\beta_i,\ e^{-b_j}\left(e^{b_j}y_j\eta_i+y_k\eta_i+e^{b_j}x_j\xi_i+e^{b_j}x_k\xi_i\right),$
$\frac{1}{2}e^{-b_j}\left(2e^{b_j}+\epsilon\, y_j y_k\eta_i^2-2e^{b_j}\epsilon\, a_j x_k\xi_i+e^{b_j}\epsilon\, x_j x_k\xi_i^2\right)\Big]$

$\mathrm{tdS}_{i\_} := \mathbb{E}\Big[-a_i\alpha_i-b_i\beta_i,\ e^{-b_i}\left(-e^{2b_i+\alpha_i}y_i\eta_i-e^{b_i+\alpha_i}x_i\xi_i-e^{b_i+\alpha_i}\eta_i\xi_i+e^{2b_i+\alpha_i}\eta_i\xi_i\right),$
$\frac{1}{4}e^{-2b_i}\Big(4e^{2b_i}+4e^{3b_i+\alpha_i}\epsilon\, y_i\eta_i-4e^{3b_i+\alpha_i}\epsilon\, y_i\beta_i\eta_i-2e^{4b_i+2\alpha_i}\epsilon\, y_i^2\eta_i^2-$
$4e^{2b_i+\alpha_i}\epsilon\, a_i x_i\xi_i-4e^{2b_i+\alpha_i}\epsilon\, x_i\beta_i\xi_i+4e^{2b_i+\alpha_i}\epsilon\,\eta_i\xi_i-4e^{3b_i+\alpha_i}\epsilon\,\eta_i\xi_i+$
$4e^{3b_i+\alpha_i}\epsilon\, a_i\eta_i\xi_i-4e^{3b_i+2\alpha_i}\epsilon\, x_i y_i\eta_i\xi_i-4e^{2b_i+\alpha_i}\epsilon\,\beta_i\eta_i\xi_i+4e^{3b_i+\alpha_i}\epsilon\,\beta_i\eta_i\xi_i-$
$2e^{3b_i+2\alpha_i}\epsilon\, y_i\eta_i^2\xi_i+6e^{4b_i+2\alpha_i}\epsilon\, y_i\eta_i^2\xi_i-2e^{2b_i+2\alpha_i}\epsilon\, x_i^2\xi_i^2-2e^{2b_i+2\alpha_i}\epsilon\, x_i\eta_i\xi_i^2+$
$6e^{3b_i+2\alpha_i}\epsilon\, x_i\eta_i\xi_i^2-e^{2b_i+2\alpha_i}\epsilon\,\eta_i^2\xi_i^2+4e^{3b_i+2\alpha_i}\epsilon\,\eta_i^2\xi_i^2-3e^{4b_i+2\alpha_i}\epsilon\,\eta_i^2\xi_i^2\Big)\Big]$

First check the double formulas on the generators agree with SL2Portfolio.pdf:

In[∘]:=
```
{
  " [a,y] = "   ((𝔼[0, 0, y₂ a₁] ~ B₁,₂ ~ tdm₁,₂→₁)⟦3⟧ - (𝔼[0, 0, y₁ a₂] ~ B₁,₂ ~ tdm₁,₂→₁)⟦3⟧),
  " [b,x] = "
   ((𝔼[0, 0, x₂ b₁] ~ B₁,₂ ~ tdm₁,₂→₁)⟦3⟧ - (𝔼[0, 0, x₁ b₂] ~ B₁,₂ ~ tdm₁,₂→₁)⟦3⟧), " xy - qyx = "
   ((𝔼[0, 0, x₁ y₂] ~ B₁,₂ ~ tdm₁,₂→₁)⟦3⟧ - (1 + ∈) (𝔼[0, 0, y₁ x₂] ~ B₁,₂ ~ tdm₁,₂→₁)⟦3⟧)
 } /. {z_1 → z, ⅇ → E} // Expand // Factor
{
  " △(a) = " ((𝔼[0, 0, a₁] ~ B₁ ~ tdΔ₁→₁,₂)⟦3⟧),
  " △(x) = " ((𝔼[0, 0, x₁] ~ B₁ ~ tdΔ₁→₁,₂)⟦3⟧),
  " △(b) = " ((𝔼[0, 0, b₁] ~ B₁ ~ tdΔ₁→₁,₂)⟦3⟧),
  " △(y) = " ((𝔼[0, 0, y₁] ~ B₁ ~ tdΔ₁→₁,₂)⟦3⟧)
 } /. {ⅇ → E} // Simplify
{
  " S(a) = " ((𝔼[0, 0, a₁] ~ B₁ ~ tdS₁)⟦3⟧),
  " S(x) = " ((𝔼[0, 0, x₁] ~ B₁ ~ tdS₁)⟦3⟧),
  " S(b) = " ((𝔼[0, 0, b₁] ~ B₁ ~ tdS₁)⟦3⟧),
  " S(y) = " ((𝔼[0, 0, y₁] ~ B₁ ~ tdS₁)⟦3⟧)
 } /. {z_1 → z, ⅇ → E} // Simplify
```

Out[∘]= $\{-\ [a,y]\ =\ y,\ \ [b,x]\ =\ x\in,\ \ xy\ -\ qyx\ =\ e^{-b}\left(-1+e^{b}+a\in\right)\}$

Out[∘]= $\{\ \triangle(a)\ =\ (a_1+a_2),\ \ \triangle(x)\ =\ \left(x_1+\left(1-\in a_1\right)x_2\right),\ \ \triangle(b)\ =\ \left(b_1+b_2\right),\ \ \triangle(y)\ =\ \left(y_1+e^{-b_1}y_2\right)\}$

Out[∘]= $\{-\ S(a)\ =\ a,\ -\ S(x)\ =\ (x+a\,x\in),\ -\ S(b)\ =\ b,\ \ S(y)\ =\ e^{b}\,y\left(-1+\in\right)\}$

Hopf algebra axioms on double

(co)-associativity

In[∘]:=
```
{(tdΔ₁→₁,₂ ~ B₂ ~ tdΔ₂→₂,₃) ≡ (tdΔ₁→₁,₃ ~ B₁ ~ tdΔ₁→₁,₂),
 (tdm₁,₂→₁ ~ B₁ ~ tdm₁,₃→₁) ≡ (tdm₂,₃→₂ ~ B₂ ~ tdm₁,₂->₁)}
```

Out[∘]= {True, True}

Δ is an algebra morphism

In[∘]:=
```
tdm₁,₂→₁ ~ B₁ ~ tdΔ₁→₁,₂ ≡ (tdΔ₁→₁,₃ tdΔ₂→₂,₄) ~ B₁,₂,₃,₄ ~ (tdm₃,₄→₂ tdm₁,₂→₁) /. ⅇ → E // Timing
```

Out[∘]= {247., True}

S is convolution inverse of id

In[∘]:=
```
{(tdΔ₁→₁,₂ ~ B₁ ~ tdS₁) ~ B₁,₂ ~ tdm₁,₂→₁, (tdΔ₁→₁,₂ ~ B₂ ~ tdS₂) ~ B₁,₂ ~ tdm₁,₂→₁} /. ⅇ → E // Timing
```

Out[∘]= {95.136, {𝔼[0, 0, 1], 𝔼[0, 0, 1]}}

S is a (co)-algebra anti-morphism

In[∘]:=
```
{tdm₁,₂→₁ ~ B₁ ~ tdS₁ ≡ (tdS₁ tdS₂) ~ B₁,₂ ~ tdm₂,₁→₁,
   tdS₁ ~ B₁ ~ tdΔ₁→₁,₂ ≡ tdΔ₁→₂,₁ ~ B₁,₂ ~ (tdS₁ tdS₂)} /. ⅇ → E // Expand // Timing
```

Out[∘]= {151.288, {True, True}}

R-matrix

In[∘]:=
```
Series[e_{q,1}[z] /. {z → y_i x_j, q → 1 + ρ}, {ρ, 0, 1}] /. {e → E, ρ → ϵ}
```

Out[∘]= $e^{x_j y_i} - \frac{1}{4}\left(e^{x_j y_i} x_j^2 y_i^2\right) \epsilon + O[\epsilon]^2$

In[∘]:=
```
e_{q_,k_}[x_] := e^( Sum_{j=1}^{k+1} (1 - q)^j x^j / (j (1 - q^j)) )
tR_{i_,j_} := 𝔼[b_i a_j, y_i x_j, 1 - ϵ 1/4 y_i^2 x_j^2] (*First two terms in Faddeev-Quesne formula*)
```

Quasi-triangular axiom 1:

In[∘]:=
```
tR_{1,2} ~ B_1 ~ tdΔ_{1->1,3} ≡ (tR_{1,4} tR_{3,2}) ~ B_{2,4} ~ tdm_{2,4→2} /. {e → E}
```

Out[∘]= True

Quasi-triangular axiom 2:

In[∘]:=
```
((tdΔ_{1->1,2} tR_{3,4}) ~ B_{1,2,3,4} ~ (tdm_{1,3→1} tdm_{2,4→2}) /. {e → E}) ≡
  ((tdΔ_{1→2,1} tR_{3,4}) ~ B_{1,2,3,4} ~ (tdm_{3,1→1} tdm_{4,2→2}) /. {e → E})
```

Out[∘]= True

Reidemeister 3:

In[∘]:=
```
((tR_{1,2} tR_{4,3} tR_{5,6}) ~ B_{1,4} ~ tdm_{1,4→1} ~ B_{2,5} ~ tdm_{2,5→2} ~ B_{3,6} ~ tdm_{3,6→3} /. e → E) ≡
   ((tR_{1,6} tR_{2,3} tR_{4,5}) ~ B_{1,4} ~ tdm_{1,4→1} ~ B_{2,5} ~ tdm_{2,5→2} ~ B_{3,6} ~ tdm_{3,6→3} /. e → E) // Timing
```

Out[∘]= {41.732, True}

In[∘]:=
```
(*Deriving tR̄ formula*)
Expand /@ tR_{i,j} ~ B_j ~ tdS_j
```

Out[∘]= $\mathbb{E}\left[-a_j b_i,\ -e^{b_i} x_j y_i,\ 1 - e^{b_i} \epsilon a_j x_j y_i - \frac{3}{4} e^{2 b_i} \epsilon x_j^2 y_i^2\right]$

In[∘]:=
```
tR̄_{i_,j_} := 𝔼[-a_j b_i, -e^{b_i} x_j y_i, 1 - e^{b_i} ϵ a_j x_j y_i - 3/4 e^{2 b_i} ϵ x_j^2 y_i^2]
```

Reidemeister 2

In[∘]:=
```
{(tR̄_{1,2} tR_{3,4}) ~ B_{1,2,3,4} ~ (tdm_{1,3→1} tdm_{2,4→2}), (tR_{1,2} tR̄_{3,4}) ~ B_{1,2,3,4} ~ (tdm_{1,3→1} tdm_{2,4→2})}
```

Out[∘]= {𝔼[0, 0, 1], 𝔼[0, 0, 1]}

Changing to the central variable t

The full zipping procedure only works when the coefficient functions commute.

We therefore need to introduce the central element $t = \epsilon a - b$.

In[∘]:=
```
b2t[i_] := 𝔼[α[i] a[i] - β[i] t[i], ξ[i] x[i] + η[i] y[i], 1 + ∈ β[i] a[i]]
t2b[i_] := 𝔼[α[i] a[i] - τ[i] b[i], ξ[i] x[i] + η[i] y[i], 1 + ∈ τ[i] a[i]]
```

In[∘]:=
```
(*converting the Hopf algebra operations*)
(t2b_i t2b_j) ~ B_{i,j} ~ tdm_{i,j→k} ~ B_k ~ b2t_k
t2b_i ~ B_i ~ tdΔ_{i→j,k} ~ B_{j,k} ~ (b2t_j b2t_k) /. ∈ → E
t2b_i ~ B_i ~ tdS_i ~ B_i ~ b2t_i
```

Out[∘]=
$$\mathbb{E}\Big[a_k\alpha_i + a_k\alpha_j + t_k\tau_i + t_k\tau_j,\ e^{-\alpha_i-\alpha_j}\left(e^{\alpha_i+\alpha_j} y_k\eta_i + e^{\alpha_j} y_k\eta_j + e^{\alpha_i} x_k\xi_i + e^{\alpha_i+\alpha_j}\eta_j\xi_i - e^{t_k+\alpha_i+\alpha_j}\eta_j\xi_i + e^{\alpha_i+\alpha_j} x_k\xi_j\right),$$
$$\frac{1}{4} e^{-\alpha_i-\alpha_j}\left(4 e^{\alpha_i+\alpha_j} + 8 e^{t_k+\alpha_i+\alpha_j}\in a_k\eta_j\xi_i + 4\in x_k y_k\eta_j\xi_i + 2 e^{\alpha_j}\in y_k\eta_j^2\xi_i - 6 e^{t_k+\alpha_j}\in y_k\eta_j^2\xi_i + 2 e^{\alpha_i}\in x_k\eta_j\xi_i^2 - 6 e^{t_k+\alpha_i}\in x_k\eta_j\xi_i^2 + e^{\alpha_i+\alpha_j}\in\eta_j^2\xi_i^2 - 4 e^{t_k+\alpha_i+\alpha_j}\in\eta_j^2\xi_i^2 + 3 e^{2t_k+\alpha_i+\alpha_j}\in\eta_j^2\xi_i^2\right)\Big]$$

Out[∘]=
$$\mathbb{E}\Big[a_j\alpha_i + a_k\alpha_i + t_j\tau_i + t_k\tau_i,\ y_j\eta_i + e^{t_j} y_k\eta_i + x_j\xi_i + x_k\xi_i,\ \frac{1}{2}\left(2 - 2 e^{t_j}\in a_j y_k\eta_i + e^{t_j}\in y_j y_k\eta_i^2 - 2\in a_j x_k\xi_i + \in x_j x_k\xi_i^2\right)\Big]$$

Out[∘]=
$$\mathbb{E}\Big[-a_i\alpha_i - t_i\tau_i,\ e^{-t_i}\left(-e^{\alpha_i} y_i\eta_i - e^{t_i+\alpha_i} x_i\xi_i + e^{\alpha_i}\eta_i\xi_i - e^{t_i+\alpha_i}\eta_i\xi_i\right),$$
$$\frac{1}{4} e^{-2t_i}\left(4 e^{2t_i} + 4 e^{t_i+\alpha_i}\in y_i\eta_i - 4 e^{t_i+\alpha_i}\in a_i y_i\eta_i - 2 e^{2\alpha_i}\in y_i^2\eta_i^2 - 4 e^{2t_i+\alpha_i}\in a_i x_i\xi_i - 4 e^{t_i+\alpha_i}\in\eta_i\xi_i + 4 e^{2t_i+\alpha_i}\in\eta_i\xi_i + 8 e^{t_i+\alpha_i}\in a_i\eta_i\xi_i - 4 e^{t_i+2\alpha_i}\in x_i y_i\eta_i\xi_i + 6 e^{2\alpha_i}\in y_i\eta_i^2\xi_i - 2 e^{t_i+2\alpha_i}\in y_i\eta_i^2\xi_i - 2 e^{2t_i+2\alpha_i}\in x_i^2\xi_i^2 + 6 e^{t_i+2\alpha_i}\in x_i\eta_i\xi_i^2 - 2 e^{2t_i+2\alpha_i}\in x_i\eta_i\xi_i^2 - 3 e^{2\alpha_i}\in\eta_i^2\xi_i^2 + 4 e^{t_i+2\alpha_i}\in\eta_i^2\xi_i^2 - e^{2t_i+2\alpha_i}\in\eta_i^2\xi_i^2\right)\Big]$$

In[∘]:=
$$\mathrm{dm}_{i\_,j\_\to k\_} := \mathbb{E}\Big[a_k\alpha_i + a_k\alpha_j + t_k\tau_i + t_k\tau_j,\ e^{-\alpha_i-\alpha_j}\left(e^{\alpha_i+\alpha_j} y_k\eta_i + e^{\alpha_j} y_k\eta_j + e^{\alpha_i} x_k\xi_i + e^{\alpha_i+\alpha_j}\eta_j\xi_i - e^{t_k+\alpha_i+\alpha_j}\eta_j\xi_i + e^{\alpha_i+\alpha_j} x_k\xi_j\right),$$
$$\frac{1}{4} e^{-\alpha_i-\alpha_j}\left(4 e^{\alpha_i+\alpha_j} + 8 e^{t_k+\alpha_i+\alpha_j}\in a_k\eta_j\xi_i + 4\in x_k y_k\eta_j\xi_i + 2 e^{\alpha_j}\in y_k\eta_j^2\xi_i - 6 e^{t_k+\alpha_j}\in y_k\eta_j^2\xi_i + 2 e^{\alpha_i}\in x_k\eta_j\xi_i^2 - 6 e^{t_k+\alpha_i}\in x_k\eta_j\xi_i^2 + e^{\alpha_i+\alpha_j}\in\eta_j^2\xi_i^2 - 4 e^{t_k+\alpha_i+\alpha_j}\in\eta_j^2\xi_i^2 + 3 e^{2t_k+\alpha_i+\alpha_j}\in\eta_j^2\xi_i^2\right)\Big]$$
$$\mathrm{d}\Delta_{i\_\to j\_,k\_} := \mathbb{E}\Big[a_j\alpha_i + a_k\alpha_i + t_j\tau_i + t_k\tau_i,\ y_j\eta_i + e^{t_j} y_k\eta_i + x_j\xi_i + x_k\xi_i,\ \frac{1}{2}\left(2 - 2 e^{t_j}\in a_j y_k\eta_i + e^{t_j}\in y_j y_k\eta_i^2 - 2\in a_j x_k\xi_i + \in x_j x_k\xi_i^2\right)\Big]$$
$$\mathrm{dS}_{i\_} := \mathbb{E}\Big[-a_i\alpha_i - t_i\tau_i,\ e^{-t_i}\left(-e^{\alpha_i} y_i\eta_i - e^{t_i+\alpha_i} x_i\xi_i + e^{\alpha_i}\eta_i\xi_i - e^{t_i+\alpha_i}\eta_i\xi_i\right),$$
$$\frac{1}{4} e^{-2t_i}\left(4 e^{2t_i} + 4 e^{t_i+\alpha_i}\in y_i\eta_i - 4 e^{t_i+\alpha_i}\in a_i y_i\eta_i - 2 e^{2\alpha_i}\in y_i^2\eta_i^2 - 4 e^{2t_i+\alpha_i}\in a_i x_i\xi_i - 4 e^{t_i+\alpha_i}\in\eta_i\xi_i + 4 e^{2t_i+\alpha_i}\in\eta_i\xi_i + 8 e^{t_i+\alpha_i}\in a_i\eta_i\xi_i - 4 e^{t_i+2\alpha_i}\in x_i y_i\eta_i\xi_i + 6 e^{2\alpha_i}\in y_i\eta_i^2\xi_i - 2 e^{t_i+2\alpha_i}\in y_i\eta_i^2\xi_i - 2 e^{2t_i+2\alpha_i}\in x_i^2\xi_i^2 + 6 e^{t_i+2\alpha_i}\in x_i\eta_i\xi_i^2 - 2 e^{2t_i+2\alpha_i}\in x_i\eta_i\xi_i^2 - 3 e^{2\alpha_i}\in\eta_i^2\xi_i^2 + 4 e^{t_i+2\alpha_i}\in\eta_i^2\xi_i^2 - e^{2t_i+2\alpha_i}\in\eta_i^2\xi_i^2\right)\Big]$$

Does it match the m-tensor from SLPortfolioProgram.nb? They do match on the generators?

```
In[•]:= {
    " [t,y] = "   ((𝔼[0, 0, y2 t1] ~B1,2~dm1,2→1) [[3]] - (𝔼[0, 0, y1 t2] ~B1,2~dm1,2→1) [[3]]),
    " [t,a] = "   ((𝔼[0, 0, a2 t1] ~B1,2~dm1,2→1) [[3]] - (𝔼[0, 0, a1 t2] ~B1,2~dm1,2→1) [[3]]),
    " [t,x] = "   ((𝔼[0, 0, x2 t1] ~B1,2~dm1,2→1) [[3]] - (𝔼[0, 0, x1 t2] ~B1,2~dm1,2→1) [[3]]),
    " [a,y] = "   ((𝔼[0, 0, y2 a1] ~B1,2~dm1,2→1) [[3]] - (𝔼[0, 0, y1 a2] ~B1,2~dm1,2→1) [[3]]),
    " [x,a] = "   ((𝔼[0, 0, a2 x1] ~B1,2~dm1,2→1) [[3]] - (𝔼[0, 0, a1 x2] ~B1,2~dm1,2→1) [[3]]),
    " xy - qyx = "
     ((𝔼[0, 0, x1 y2] ~B1,2~dm1,2→1) [[3]] - (1 + ϵ) (𝔼[0, 0, y1 x2] ~B1,2~dm1,2→1) [[3]])
   } /. {z_1 → z, ⅇ → E} // Expand // Factor
  {
   " Δ(a) = " ((𝔼[0, 0, a1] ~B1~dΔ1→1,2) [[3]]),
   " Δ(x) = " ((𝔼[0, 0, x1] ~B1~dΔ1→1,2) [[3]]),
   " Δ(t) = " ((𝔼[0, 0, t1] ~B1~dΔ1→1,2) [[3]]),
   " Δ(y) = " ((𝔼[0, 0, y1] ~B1~dΔ1→1,2) [[3]])
  } /. {ⅇ → E} // Simplify
  {
   " S(a) = " ((𝔼[0, 0, a1] ~B1~dS1) [[3]]),
   " S(x) = " ((𝔼[0, 0, x1] ~B1~dS1) [[3]]),
   " S(t) = " ((𝔼[0, 0, t1] ~B1~dS1) [[3]]),
   " S(y) = " ((𝔼[0, 0, y1] ~B1~dS1) [[3]])
  } /. {z_1 → z, ⅇ → E} // Simplify
```

```
Out[•]= {0, 0, 0, - [a,y] = y, - [x,a] = x,  xy - qyx = (1 - ⅇ^t + 2 a ⅇ^t ϵ)}
Out[•]= { Δ(a) = (a1 + a2),  Δ(x) = (x1 + (1 - ϵ a1) x2),
          Δ(t) = (t1 + t2),  Δ(y) = (y1 - ⅇ^t1 (-1 + ϵ a1) y2)}
Out[•]= {- S(a) = a, - S(x) = (x + a x ϵ), - S(t) = t, - S(y) = ⅇ^-t y (1 + (-1 + a) ϵ)}
```

Hopf algebra axioms on double

(co)-associativity

```
In[•]:= {(dΔ1→1,2~B2~dΔ2→2,3) ≡ (dΔ1→1,3~B1~dΔ1→1,2),
        (dm1,2→1~B1~dm1,3→1) ≡ (dm2,3→2~B2~dm1,2->1)}
Out[•]= {True, True}
```

Δ is an algebra morphism

```
In[•]:= dm1,2→1~B1~dΔ1→1,2 ≡ (dΔ1→1,3 dΔ2→2,4) ~B1,2,3,4~ (dm3,4→2 dm1,2→1) /. ⅇ → E // Timing
Out[•]= {105.488, True}
```

S is convolution inverse of id

```
In[•]:= {(dΔ1→1,2~B1~dS1) ~B1,2~dm1,2→1, (dΔ1→1,2~B2~dS2) ~B1,2~dm1,2→1} /. ⅇ → E // Timing
Out[•]= {34.684, {𝔼[0, 0, 1], 𝔼[0, 0, 1]}}
```

S is a (co)-algebra anti-morphism

In[∘]:= $\{dm_{1,2\to1}\sim B_1\sim dS_1 \equiv (dS_1\, dS_2)\sim B_{1,2}\sim dm_{2,1\to1},\ dS_1\sim B_1\sim d\Delta_{1\to1,2} \equiv d\Delta_{1\to2,1}\sim B_{1,2}\sim (dS_1\, dS_2)\}$ /. e → E // Expand // Timing

Out[∘]= {133.812, {True, True}}

R-matrix

In[∘]:= $tR_{i,j}\sim B_{i,j}\sim (b2t_i\, b2t_j)$
$\overline{tR}_{i,j}\sim B_{i,j}\sim (b2t_i\, b2t_j)$

Out[∘]= $\mathbb{E}\left[-a_j t_i,\ x_j y_i,\ \frac{1}{4}\left(4+4\in a_i a_j-\in x_j^2 y_i^2\right)\right]$

Out[∘]= $\mathbb{E}\left[a_j t_i,\ -e^{-t_i} x_j y_i,\ \frac{1}{4} e^{-2 t_i}\left(4 e^{2 t_i}-4 e^{2 t_i}\in a_i a_j-4 e^{t_i}\in a_i x_j y_i-4 e^{t_i}\in a_j x_j y_i-3\in x_j^2 y_i^2\right)\right]$

In[∘]:= (*Deriving $\overline{R}$ formula*)
Expand /@ $R_{i,j}\sim B_j\sim dS_j$

Out[∘]= $\mathbb{E}\left[a_j t_i,\ -e^{-t_i} x_j y_i,\ 1-\in a_i a_j-e^{-t_i}\in a_i x_j y_i-e^{-t_i}\in a_j x_j y_i-\frac{3}{4} e^{-2 t_i}\in x_j^2 y_i^2\right]$

$R_{i\_,j\_} := \mathbb{E}\left[-a_j t_i,\ x_j y_i,\ \frac{1}{4}\left(4+4\in a_i a_j-\in x_j^2 y_i^2\right)\right]$
$\overline{R}_{i\_,j\_} :=$
$\mathbb{E}\left[a_j t_i,\ -e^{-t_i} x_j y_i,\ \frac{1}{4} e^{-2 t_i}\left(4 e^{2 t_i}-4 e^{2 t_i}\in a_i a_j-4 e^{t_i}\in a_i x_j y_i-4 e^{t_i}\in a_j x_j y_i-3\in x_j^2 y_i^2\right)\right]$

Quasi-triangular axiom 1:

In[∘]:= $R_{1,2}\sim B_1\sim d\Delta_{1\text{->}1,3} \equiv (R_{1,4}\, R_{3,2})\sim B_{2,4}\sim dm_{2,4\to2}$

Out[∘]= True

Quasi-triangular axiom 2:

In[∘]:= $(d\Delta_{1\text{->}1,2}\, R_{3,4})\sim B_{1,2,3,4}\sim (dm_{1,3\to1}\, dm_{2,4\to2}) \equiv (d\Delta_{1\to2,1}\, R_{3,4})\sim B_{1,2,3,4}\sim (dm_{3,1\to1}\, dm_{4,2\to2})$

Out[∘]= True

Reidemeister 3:

In[∘]:= $((R_{1,2}\, R_{4,3}\, R_{5,6})\sim B_{1,4}\sim dm_{1,4\to1}\sim B_{2,5}\sim dm_{2,5\to2}\sim B_{3,6}\sim dm_{3,6\to3} \equiv$
$(R_{1,6}\, R_{2,3}\, R_{4,5})\sim B_{1,4}\sim dm_{1,4\to1}\sim B_{2,5}\sim dm_{2,5\to2}\sim B_{3,6}\sim dm_{3,6\to3})$ /. e → E // Timing

Out[∘]= {36.072, True}

Deriving the Drinfeld element u and its inverse ui

In[∘]:= $u_{i\_} := R_{1,2}\sim B_1\sim dS_1\sim B_{1,2}\sim dm_{2,1\to i}$
$ui_{i\_} := R_{1,2}\sim B_2\sim dS_2\sim B_2\sim dS_2\sim B_{1,2}\sim dm_{2,1\to i}$

In[∘]:= {u_1, ui_1}

Out[∘]= $\{\mathbb{E}[a_1 t_1, -e^{-t_1} x_1 y_1, \frac{1}{4} e^{-t_1} (4 e^{2 t_1} - 8 e^{2 t_1} \epsilon a_1 - 4 e^{2 t_1} \epsilon a_1^2 - 4 e^{t_1} \epsilon x_1 y_1 - 8 e^{t_1} \epsilon a_1 x_1 y_1 - 3 \epsilon x_1^2 y_1^2)], \mathbb{E}[-a_1 t_1, x_1 y_1, \frac{1}{4} e^{-t_1} (4 + 8 \epsilon a_1 + 4 \epsilon a_1^2 - 4 \epsilon x_1 y_1 - \epsilon x_1^2 y_1^2)]\}$

u and ui are inverses

In[∘]:= $(u_1\, ui_2) \sim B_{1,2} \sim dm_{1,2\to 1}$

Out[∘]= $\mathbb{E}[0, 0, 1]$

The ribbon element v satisfies $v^2 = S(u)\, u$. The spinner C=uv^{-1}.
It is convenient to compute $z = S(u)\, u^{-1}$ which is something easy.

In[∘]:= $((u_1 \sim B_1 \sim dS_1)\, ui_2) \sim B_{1,2} \sim dm_{1,2\to 1}$

Out[∘]= $\mathbb{E}[0, 0, e^{-t_1} (1 + 2 \epsilon a_1)]$

So in our case S(u) = u z so S(u)u = $u^2 z$ and v = $uz^{\frac{1}{2}}$ and finally C = $uv^{-1} = z^{-\frac{1}{2}} = e^{t_1/2}(1 - \epsilon a_1)$.

$CC_{i\_} := \mathbb{E}[0, 0, e^{\frac{1}{2} t_i} (1 - \epsilon a_i)]$
$\overline{CC}_{i\_} := \mathbb{E}[0, 0, e^{\frac{-1}{2} t_i} (1 + \epsilon a_i)]$

Kinks

In[∘]:= $k1 = (R_{1,3}\, \overline{CC}_2) \sim B_{1,2} \sim dm_{1,2\to 1} \sim B_{1,3} \sim dm_{1,3\to i}$ /. $e \to E$;
$k2 = (R_{3,1}\, CC_2) \sim B_{1,2} \sim dm_{1,2\to 1} \sim B_{1,3} \sim dm_{1,3\to i}$ /. $e \to E$;
$k3 = (\overline{R}_{1,3}\, CC_2) \sim B_{1,2} \sim dm_{1,2\to 1} \sim B_{1,3} \sim dm_{1,3\to j}$ /. $e \to E$;
$k4 = (\overline{R}_{3,1}\, \overline{CC}_2) \sim B_{1,2} \sim dm_{1,2\to 1} \sim B_{1,3} \sim dm_{1,3\to j}$ /. $e \to E$;
$\{k1 \equiv k2, k3 \equiv k4, (k1\, k3) \sim B_{i,j} \sim dm_{i,j\to 1}\}$
k1
k3

Out[∘]= {True, True, $\mathbb{E}[0, 0, 1]$}

Out[∘]= $\mathbb{E}[-a_i t_i, x_i y_i, \frac{1}{4} e^{-\frac{t_i}{2}} (4 + 4 \epsilon a_i + 4 \epsilon a_i^2 - \epsilon x_i^2 y_i^2)]$

Out[∘]= $\mathbb{E}[a_j t_j, -e^{-t_j} x_j y_j, \frac{1}{4} e^{-2 t_j} (4 e^{\frac{5 t_j}{2}} - 4 e^{\frac{5 t_j}{2}} \epsilon a_j - 4 e^{\frac{5 t_j}{2}} \epsilon a_j^2 - 8 e^{\frac{3 t_j}{2}} \epsilon a_j x_j y_j - 3 e^{\frac{t_j}{2}} \epsilon x_j^2 y_j^2)]$

In[∘]:= $Kink_{i\_} := \mathbb{E}[-a_i t_i, x_i y_i, \frac{1}{4} e^{-\frac{t_i}{2}} (4 + 4 \epsilon a_i + 4 \epsilon a_i^2 - \epsilon x_i^2 y_i^2)]$
$\overline{Kink}_{j\_} :=$
$\mathbb{E}[a_j t_j, -e^{-t_j} x_j y_j, \frac{1}{4} e^{-2 t_j} (4 e^{\frac{5 t_j}{2}} - 4 e^{\frac{5 t_j}{2}} \epsilon a_j - 4 e^{\frac{5 t_j}{2}} \epsilon a_j^2 - 8 e^{\frac{3 t_j}{2}} \epsilon a_j x_j y_j - 3 e^{\frac{t_j}{2}} \epsilon x_j^2 y_j^2)]$

Reidemeister 2:

In[∘]:= $\left(R_{1,2}\,\overline{R}_{3,4}\right)\sim B_{1,3}\sim dm_{1,3\to 1}\sim B_{2,4}\sim dm_{2,4\to 2}$

Out[∘]= $\mathbb{E}[0, 0, 1]$

cyclic Reidemeister 2:

In[∘]:= $\left(R_{1,4}\,\overline{R}_{5,2}\,\overline{CC}_3\right)\sim B_{2,4}\sim dm_{2,4\to 2}\sim B_{1,3}\sim dm_{1,3\to 1}\sim B_{1,5}\sim dm_{1,5\to 1} \equiv \overline{CC}_1$

Out[∘]= True

Trefoil

In[∘]:=
```
Z = R_{1,5} R_{6,2} R_{3,7} CC̄_4 Kink̄_8 Kink̄_9 Kink̄_10;
Do[Print["doing ", r]; Z = Z ~ B_{1,r} ~ dm_{1,r→1} /. ε → E, {r, 2, 10}] // Timing
Z
1 / (Z[[3]] /. e^(c_. t1) → T^c) // Expand
```

doing 2

doing 3

doing 4

doing 5

doing 6

doing 7

doing 8

doing 9

doing 10

Out[∘]= {867.924, Null}

Out[∘]= $\mathbb{E}\Big[0, 0, \big(e^{t_1} - 2e^{2t_1} + 3e^{3t_1} - 2e^{4t_1} + e^{5t_1} - e^{2t_1}\epsilon + 2e^{3t_1}\epsilon - 3e^{4t_1}\epsilon + 2e^{5t_1}\epsilon - 2e^{t_1}\epsilon a_1 + 2e^{2t_1}\epsilon a_1 - 2e^{4t_1}\epsilon a_1 + 2e^{5t_1}\epsilon a_1 - 2e^{t_1}\epsilon x_1 y_1 - 2e^{4t_1}\epsilon x_1 y_1\big) \big/ \big(1 - 3e^{t_1} + 6e^{2t_1} - 7e^{3t_1} + 6e^{4t_1} - 3e^{5t_1} + e^{6t_1}\big)\Big]$

In[∘]:=
```
1 / Z[[3]] /. {e^(c_. t1) → T^c, ε → 0} // Factor // Expand
Coefficient[Z[[3]] /. e^(c_. t1) → T^c, ε] // Factor
```

Out[∘]= $-1 + \frac{1}{T} + T$

Out[∘]= $\left(T\left(-T + 2T^2 - 3T^3 + 2T^4 - 2a_1 + 2Ta_1 - 2T^3a_1 + 2T^4a_1 - 2x_1y_1 - 2T^3x_1y_1\right)\right) \big/ \left(1 - T + T^2\right)^3$

In[∘]:= `Z[[3]] /. e^(c_. t1) → T^c // Factor`

Out[∘]= $\left(T\left(1 - 2T + 3T^2 - 2T^3 + T^4 - T\epsilon + 2T^2\epsilon - 3T^3\epsilon + 2T^4\epsilon - 2\epsilon a_1 + 2T\epsilon a_1 - 2T^3\epsilon a_1 + 2T^4\epsilon a_1 - 2\epsilon x_1y_1 - 2T^3\epsilon x_1y_1\right)\right) \big/ \left(1 - T + T^2\right)^3$